A SAS Macro to Calculate the Ankle-Brachial Index

Rita A. Dale, Colorado Prevention Center, Denver, Colorado

ABSTRACT

The Ankle-Brachial Index (ABI) is used to diagnose peripheral arterial disease (PAD). The ABI is a ratio of systolic blood pressures obtained from the brachial artery in the arm and the dorsalis pedis (DP) or posterior tibial (PT) arteries in the ankle. Multiple pressures may be taken at each artery on both the right and left sides of the body to increase the precision of the measurement. As a result the statistician can have 6, 12 or even 18 pressures to calculate the ratio. It can be easy to make cut and paste errors when developing programs that may have similar variable names, but fortunately the naming convention lends itself to macro programming. The SAS® macro presented here was developed to decrease the chance for error and to reduce programming time necessary to calculate the ABI for various clinical protocols.

INTRODUCTION

Peripheral Arterial Disease (PAD) is a disease characterized by arterial stenosis in the legs resulting in pain upon walking. The Ankle-Brachial Index (ABI) is a measure of impaired blood flow used to diagnose PAD. In PAD studies the value of the ABI is used as both inclusion criteria to the study and as a follow-up measure. Therefore, the ABI must be calculated in the field to determine inclusion status of the subject. Often a detailed worksheet is supplied to the study site to assist in the calculation. However, for data analysis the statistician will calculate the ABI using the raw systolic blood pressures recorded on the case report forms to ensure consistent and correct calculations.

The ABI is a ratio of systolic blood pressures obtained from the brachial artery in the right or left arm and either the dorsalis pedis (DP) or posterior tibial (PT) artery in the right or left ankle. The approximate location of these six arteries is labeled in Figure 1. The calculation for the ABI, though somewhat protocol dependent, is usually derived from the maximum of the DP or PT ankle pressures per leg divided by the maximum of two brachial pressures regardless of side. Then the minimum of the left and right leg ratios is used as the diagnostic ABI value. An ABI of 0.90 or lower indicates a diagnosis of PAD. The standard equation is included at the top of Figure 1.

Figure 1.

\[ ABI = \min \left( \frac{\max(\text{right } DP, \text{right } PT)}{\max(\text{right } ARM, \text{left } ARM)} \right) \left( \frac{\max(\text{left } DP, \text{left } PT)}{\max(\text{right } ARM, \text{left } ARM)} \right) \]

Right side

Brachial (ARM)
Posterior Tibial (PT)
Dorsalis Pedis (DP)

Left side

Brachial (ARM)
Posterior Tibial (PT)
Dorsalis Pedis (DP)
%rl:  THE MACRO WITHIN THE %abi MACRO

Sometimes only a single pressure is taken at each of the six arteries. However, in industry standard clinical trials multiple pressures are usually taken at each artery to increase precision. Then the average of those pressures is used in the calculation. The %abi macro was written to allow for cases when only one pressure is taken at each artery and when multiple pressures are taken at each artery. This was done by placing a macro, referred to as the %rl macro, within the %abi macro. Conditional macro programming is used in the %rl macro to either rename the variables from each artery or to create a summary measure of multiple pressures from each artery. This must be called for each of the six arteries and will result in six new variables using the &art&side naming convention. For the %abi macro to run properly the parameter &art should be named for the artery (arm, dp, or pt) and the parameter &side should be named for the side of the body (right or left). When multiple pressures are taken per artery %rl uses a variable list to calculate a summary measure for the multiple pressures per artery. This summary measure, &artsum, is typically the average of the pressures; however the macro allows for any function of the replicate pressures. In the case where only one pressure is taken at each artery the %rl macro simply renames the variables from each artery using the &art&side naming convention. The parameters &lastmeas and &artsum are omitted when a single pressure is taken per artery. The sample calls included in this paper show the case when more than one pressure is taken per artery and the mean is called as the summary function.

```sas
%macro rl (art,side,frstmeas,lastmeas,artsum);
  %if &lastmeas eq %then &art&side=&frstmeas;
  %else &art&side=&artsum(of &frstmeas-&lastmeas);
%mend rl;

%rl (arm,right,arm_r1,arm_r2,mean);
%rl (arm,left, arm_l1,arm_l2,mean);
%rl (dp, right,dpp_r1,dpp_r2,mean);
%rl (dp, left, dpp_l1,dpp_l2,mean);
%rl (pt, right,ptp_r1,ptp_r2,mean);
%rl (pt, left, ptp_l1,ptp_l2,mean);
```

The variable list used when more than one pressure is taken per artery requires the macro variables &frstmeas and &lastmeas. These variables must meet the naming criteria of SAS variable lists. Criteria for SAS variables lists are well defined in SAS help and include number range lists, name range lists, name prefix lists, and special SAS name lists. Replicate pressures per artery will often have sensible variables names ending with a _1, _2 thus fitting the number range list criteria. However, if the variable names do not meet the SAS variable list criteria they must be renamed prior to using the macro.

The %rl macro is named as such to symbolize the right (r) and left (l) sides of the body. Within the six required program calls for the %rl macro is the only place where the programmer has to enter the variables for the right and left side of the body. This is a huge advantage to the programmer because similarly named variables, as we would expect these to be, can often be interchanged or cut and pasted in error when programming. Errors made when referring to right and left sided measurements was actually the primary motivation for developing this macro.

%abi:  THE MAIN MACRO

The six pressures resulting from the %rl macro are sensibly named as ‘armright’, ‘armleft’, ‘dpright’, ‘dpleft’, ‘ptright’, and ‘ptleft’ when called as specified above. These variables are then used to calculate the ABI. Since the ABI calculation is not entirely standardized and can be protocol dependent the macro allows the programmer to choose the function (e.g. min, max, median, or mean) of the pressures needed for the calculation. The macro variable &brachsum allows the programmer to select any function of the left and right brachial (arm) pressures. Both the left and right ankle summary pressures select between the DP and the PT pressure with the function &anksum. The ABI is then calculated as the ratio of the selected ankle and brachial pressure for each leg (leftabi and rightabi). The final ABI is selected with the function &abisum. The sample code included calls these macro variables with the standard calculation: &brachsum=maximum of right or left brachial pressure, &anksum=maximum of DP or PT ankle pressures per side, and &abisum=minimum of right or left leg ABI.
CONCLUSION

Statisticians and statistical programmers working on industry standard clinical trials are expected to report results with a high level of accuracy. Often they may spend an inordinate amount of time on fairly simple tasks to meet the expected standards. By incorporating the ABI calculation into a macro this particular task can be generalized across several protocols and database structures. Since the macro can be used in so many situations the chance for error and programming time are reduced. As a result the ABI can be reported with a high level of accuracy and the statistician has more time to spend on other aspects of the project.

REFERENCES


ACKNOWLEDGEMENTS

I would like to thank David Ryerson and Darryl Perry who reviewed this macro and gave me some helpful suggestions.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:
Rita A. Dale
Colorado Prevention Center
789 Sherman St. Ste. 200
Denver, CO 80203
303-860-9900 Ph.
303-860-1288 Fx.
Rita.Dale@cpcmed.org
www.cpcmed.org

SAS and all other SAS Institute Inc. products or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.
APPENDIX 1. COMPLETE %abi MACRO WITH SAMPLE MACRO CALLS

/*%abi parameters:
   bpdata = input dataset
   abidata = output dataset
   brachsum = summary function of left and right arm pressures
   anksum = summary function of left and right DP or PT ankle pressures
   abism = summary function of left or right abi */
%macro abi (bpdata,abicalc,brachsum,anksum,abism);
data &abicalc;
set &bpdata;

/*%rl parameters:
   art = the artery (arm, dp, or pt)
   side = the side of body (right or left)
   frstmeas = variable name for the 1st or only pressure taken at the
            artery
   lastmeas = variable name for the last pressure taken at the artery
     (MUST BE IN SAS VARIABLE LIST FORMAT) leave empty when 1
     pressure taken per artery
   artsum = summary function of multiple pressures per artery. Leave
     empty when 1 pressure taken per artery*/
%macro rl (art,side,frstmeas,lastmeas,artsum);
   %if &lastmeas eq %then &art&side=&frstmeas;
   %else &art&side=&artsum(of &frstmeas-&lastmeas);
%mend rl;

%rl (arm,right,armp_r1,armp_r2,mean);
%rl (arm,left, armp_l1,armp_l2,mean);
%rl (dp, right,dpp_r1,dpp_r2,mean);
%rl (dp, left, dpp_l1,dpp_l2,mean);
%rl (pt, right,ptp_r1,ptp_r2,mean);
%rl (pt, left, ptp_l1,ptp_l2,mean);

/*calculate ABI*/
/*1. use a summary measure of the right and left brachial pressures*/
brachial=&brachsum(armright,armleft);
/*use a summary measure of the left and right dp and pt pressures*/
rightankle=&anksum(dpright,ptright);
leftankle=&anksum(dpleft,ptleft);
/*right and right ankle:brachial index*/
rightabi=rightankle/brachial;
leftabi=leftankle/brachial;
/*report the indicated summary measure of the right and left abi*/
abi=&abism(leftabi,rightabi);
run;

%mend abi;

%abi (dataout,abiout,max,max,min);